FORMERLY THE CURSOR GROUP

THE BASIC EXPRESS

JOURNAL FOR THE BALLY HOME COMPUTER

VOLUMN III | APRIL | ISSUE I
$9.75 SEMI-ANNUALLY | 1981 | $2.00 PER ISSUE

WHAT'S NEW

BY

FRED CORNETT

As you can see, we've had a name change, and that "ain't" all! We've increased our issue size by fifty percent. This size increase expands the number of programs and tutorials to double what we could previously cram into an issue, and all that at no additional increase in subscription price!!

Why the name change? If the Zgrass 32 Add-under becomes the total success we think it will, we are planning to support that unit with a full size magazine (100+ pages).

As time has passed, we have found our business leaning toward diverse product development with little or no way to differentiate between them. Hence the change in name for this newsletter.

The New BASIC cartridge we discussed last month has once again been improved, and it is absolutely delightful. Jay Fenton (the man who wrote BALLY BASIC) has outdone himself! Jay changed the EDITOR function to one keystroke, and it makes programming much much easier. Many other goodies have been added, and we will discuss these changes in subsequent issues. The New BASIC should be available in about 10 weeks.



DAILY DILEMMA!!

ASTROVISION's newest game cartridge- "GALACTIC INVASION" should be available in about one month. Like the New BASIC, this cartridge has been redone about 10 times. The results are well worth waiting for. *FANTASTIC!!* This is a home version of the penny arcade game "GALAXIAN". ASTROVISION has made it a 4 player game with software selectable levels of difficulty (1-9). The cartridge is, in our opinion, better than the professional version!

The professional "GALAXIAN" level of difficulty is permently set at about the five level, whereas the home version can be set far more difficult!

The game is set up somewhat similiar to Space Invaders, with invaders in rows & files, and that is where the similarity ends. The enemy looks like an animated fly, they peel off in groups of three and do strafing runs on your position. Great sound, requires skill. *You'll love it!!*

Work is progressing on the add-under. Various changes in design have slowed up production plans. We at CURSOR keep pushing for CPM compatibility (which it

MAIL ADDRESS: P. O. BOX 266, NORTH HOLLYWOOD, CA. 91603
BUSINESS ADDRESS:1010 A. WEST MAGNOLIA BLVD. • SUITE 201 • BURBANK, CA. 91506 • (213) 843-7332

will have), and this type of change slows things down. Field test units are promised us by middle of June. ASTRO-VISION refuses to come out too early with an untested product. Better to shove a delivery date ahead a bit and truly produce a quality product!

We have received many calls beggins us for a location where Bally owners can get a *fast* return of their Bally for repairs. We've been repairing units in Southern California for the past few months. People in other parts of the country have suggested we handle mail order repairs for them as well. So--- We'll give it a try!

For a $25. fee (plus $5 return shipping) we will bond custom made heat sinks on the Input/Output chip, Custom Data chip, and Address Chip; modify your RF shielding to further reduce heat build-up; remove the ON/OFF switch from the circuit (this switch causes untold problems); tune the RF modulator to improve your television picture; and make any general repairs as necessary (does not include major parts). Most units we have repaired have not required additional parts. If your unit requires extensive repairs not covered by the $25. fee, we will ask your permission to complete the work. If by some chance we are unable to repair your unit, there will be no charge other than the $5. return postage.

We have performed the $25. General Maintenance on several hundred units and have eliminated most heat related problems, such as keyboard lockup; garbage on your TV screen; system crashes; crashes in the middle of loading an audio tape; unit refuses to start-up properly, etc..

Your Bally unit will be returned to you within 10 days from the time we receive it!

For the do-it-yourself fan, we have finally received custom IC heat sinks (to fit 40 pin) that we had made for us. We couldn't find these little beasts anywhere! By bonding these sinks on the 3 custom chips, the heat dissipation is increased vastly! *Order your set of 3 today at $3.95 including shipping & handling. CA residents add 6% tax.*

CORRECTION: A good rule to follow when programming is to avoid using the letter variables O and S. It is very easy to mistake the letter O for a numberic zero or the letter S for a numeric five. This is exactly what happened in the "HAMURABI" program on page 95 March 81 issue. We mistook the letter O for numeric Ø and slashed all the letter O positions thereby making them numeric zero. Please change the following line to read as shown below:

```
1ØØ IF W=1INPUT "WHAT IS YOUR OFFER?"O;I
    F E-O<RND (3)D=D+O;PRINT "FEED YOUR
    PEOPLEç",#1,D,"ςBUSHELS PER YEAR";W=
    Ø
```

THE BEAUTY OF THE LOOP

BY

DANIEL J. DRESCHER

If you await the arrival of each months CURSOR as I do, you're probably aware of the fascinating things being done with machine language. It seems that through the organization of THE CURSOR GROUP and their subscribers this relatively inexpensive machine that most of us bought as a toy is being transformed into a versatile and powerful little computor. However, in the midst of all this, we should keep in mind that BASIC, our computor's first language, also has many powerful commands that we should completely understand and learn to use to their fullest advantage. One of these, the topic here, is the FOR, NEXT loop.

Simply stated, the loop allows us to pick a point in the main body of a program and perform any number of instructions any number of times before going on with execution of the main body program. At the same time, it increments or decrements a variable of our choice by any amount set by the STEP value.

For example:

```
 5 CLEAR
1Ø FOR X=4 TO 64 STEP 4;BOX Ø,Ø,X,X,3;N
   EXT X
```

This program sets the value of X to 4, draws the 4 X 4 box in the center of the screen then goes back to the FOR instruction where it gets another value for X and does the whole thing over again. As you can see, the effect is

very interesting. Now add this one line which contains another loop and you will see a wild graphic effect.

```
2Ø FOR X=64 TO 4 STEP -4;BOX Ø,Ø,X,X,3;
   NEXT X;GOTO 1Ø
```

The step value can be any amount positive or negative. The variable can have any starting and ending value we wish. The use of the changing variable is only limited by our imagination. Here is a program where it is used not only to change the size of a box but also it's location.

```
 5 CLEAR;BC=Ø;FC=79
1Ø FOR X=7Ø TO 2 STEP -2
2Ø BOX -X,Ø,3,X+2,1;BOX X,Ø,3,X+2,1;NEX
   T X
```

That's one way to make laser cannons for your next space adventure!

Any number of loops can be combined one within the other. Say we have an instruction we want to execute eight times for every single execution of another instruction. For instance, a game board of some sort or some data format. These two little loops will do it.

```
 5 CLEAR
1Ø FOR Y=-35 TO 35 STEP 1Ø
2Ø FOR X=-35 TO 35 STEP 1Ø
3Ø BOX X,Y,8,8,1
4Ø NEXT X
5Ø NEXT Y
```

The loop can be used to create many interesting graphic and sound effects. It can be inserted as a time delay or used with the string arrays to manipulate data. It's beauty is in it's efficiency, and it's effective use can be seen through out the pages of CURSOR. One of the nicest was THE CHICAGO LOOP by Mike Peace in the March 80 issue. If you overlooked running that one you missed a truly creative piece of programming. The more experience one gets at working with loops and using them to control variables the more uses one will find for them.

THERE IS AN INCESSANT INFLUX OF NOVELTY IN THE WORLD AND YET WE TOLERATE INCREDIBLE DULLNESS..

-H.D. Thoreau (1817-1861)

DARTS

BY

BILL MEAD

"DARTS" is a 4 player simulation. An attempt has been made to simulate realism. Player may select 1 of 3 different throws:

1. HARD THROW (somewhat erratic). With this mode, any score is possible, 1Ø 2Ø-3Ø-4Ø. Bullseye or a complete miss.
2. MODERATE THROW (controlled). Each throw will receive a score of 1Ø-2Ø-3Ø or 4Ø. No possibility of a bullseye or complete miss.
3. VERY HARD THROW. Either a bullseye or a complete miss.

A winning score is 500. Move Joystick left or right to select throw type. Pull trigger to throw dart. Upon completion of game, press any key to begin anew.

```
 1 .DARTS
 2 .
 5 NT=1
1Ø CLEAR ;PRINT "ζ←JOYSTICK→ζCHOOSES THRO
   W.
15 PRINT ;PRINT "ζTRIGGER THROWS DART.
2Ø PRINT ;PRINT "ζζ1.ζANYTHING
3Ø PRINT ;PRINT "ζζ2.ζ1Ø,2Ø,3Ø OR 4Ø POIN
   TS
4Ø PRINT ;PRINT "ζζ3.ζBULLSEYE OR ZERO
5Ø FOR I=1TO 2ØØØ;NEXT I
6Ø GOTO 7ØØ
1ØØ CY=-1Ø;CX=14;PRINT "ζζζζζζζζζζ",
11Ø BOX 3Ø,2Ø,45,45,1
12Ø BOX 3Ø,2Ø,35,35,2
13Ø BOX 3Ø,2Ø,25,25,1
14Ø BOX 3Ø,2Ø,15,15,2
15Ø BOX 3Ø,2Ø,5,5,1;RETURN
2ØØ IF S=3GOTO 26Ø
21Ø IF S=2GOTO 28Ø
22Ø A=RND (2)
23Ø CY=-1Ø;CX=14
24Ø IF A=1PRINT "ζA MISSζ",;R=Ø;GOTO 61Ø
25Ø BOX 3Ø,2Ø,1,1,3;PRINT "BULLSEYE",;NT=3
   ;FOR J=1TO 15;MU=77;NEXT J
255 R=5Ø;@(Z)=@(Z)+R;GOTO 61Ø
26Ø B=RND (3)
27Ø IF B=1GOTO 22Ø
28Ø R=1Øx RND (4)
29Ø GOSUB 3ØØ+(Rx1Ø-1ØØ)
295 CX=Ø;CY=-1Ø;PRINT R,;@(Z)=@(Z)+R
296 NT=4;FOR J=1TO R÷1Ø;MU=74;NEXT J;NT=
   3;GOTO 61Ø
3ØØ C=RND (7);GOTO 3ØØ+C
```

```
3Ø1 BOX 1Ø,25,3,3,3;RETURN
3Ø2 BOX 1Ø,5,3,3,3;RETURN
3Ø3 BOX 19,4Ø,3,3,3;RETURN
3Ø4 BOX 5Ø,4Ø,3,3,3;RETURN
3Ø5 BOX 5Ø,14,3,3,3;RETURN
3Ø6 BOX 4Ø,Ø,3,3,3;RETURN
3Ø7 BOX 18,Ø,3,3,3;RETURN
4ØØ C=RND (5);GOTO 4ØØ+C
4Ø1 BOX 35,5,3,3,3;RETURN
4Ø2 BOX 45,12,3,3,3;RETURN
4Ø3 BOX 45,3Ø,3,3,3;RETURN
4Ø4 BOX 16,35,3,3,3;RETURN
4Ø5 BOX 15,22,3,3,3;RETURN
5ØØ C=RND (4);GOTO 5ØØ+C
5Ø1 BOX 2Ø,24,3,3,3;RETURN
5Ø2 BOX 34,1Ø,3,3,3;RETURN
5Ø3 BOX 4Ø,18,3,3,3;RETURN
5Ø4 BOX 29,3Ø,3,3,3;RETURN
6ØØ C=RND (3);GOTO 6ØØ+C
6Ø1 BOX 25,23,3,3,3;RETURN
6Ø2 BOX 33,25,3,3,3;RETURN
6Ø3 BOX 35,18,3,3,3;RETURN
61Ø NT=Ø;CY=2Ø;FOR I=1TO P
62Ø CY=CY-1Ø;CX=-70;PRINT "#",#Ø,I,"=",#
    4,@(I),
63Ø IF @(I)>499GOTO 65Ø
64Ø NEXT I;RETURN
65Ø CY=-35;CX=-7Ø;PRINT "THE WINNER IS P
    LAYERζ",#Ø,I;A=KP;RUN
7ØØ CLEAR ;NT=1;BC=125
71Ø FOR I=1TO 4;@(I)=Ø;NEXT I
72Ø INPUT "ζ#TO PLAY:"P
721 IF (P>4)+(P<1)CLEAR ;GOTO 72Ø
722 CLEAR
725 FOR Z=1TO P
726 NT=Ø;CX=-65;CY=4Ø;PRINT "PLAYERζ",#Ø
    ,Z
727 NT=12;FOR J=1TO Z;MU=72;NEXT J;NT=Ø
73Ø CX=-65;CY=3Ø;PRINT "1ζζζ2ζζζ3"
74Ø GOSUB 1ØØ
75Ø CY=22;CX=4-Sx23;PRINT "↑",
76Ø FOR I=1TO 5Ø;NEXT I;BOX CX-7,CY,5,7,
    2;IF TR(Z)GOSUB 2ØØ;GOTO 8ØØ
77Ø S=S-JX(Z);IF S<1S=1
78Ø IF S>3S=3
79Ø NT=Ø;GOTO 75Ø
8ØØ FOR I=1TO 1ØØØ;NEXT I;NEXT Z;GOTO 72
    5
```

IMPORTANT NOTE

It is sometimes difficult to ascertain the correct number of spaces with a PRINT statement. To facilitate ease of input, we are using the special character "ζ" to designate a SPACE where-ever confusion could exist.

BACK ISSUES AVAILABLE

We have had numerous requests for info regarding BACK ISSUES! The following are available:

1. Jan. 80 Contains: Electric Bill Analysis; Plastic Puzzle, Instructions for adding a Full-sized ASCII Keyboard; Life Systhesis Model.
2. Feb. 80 Contains: PEEK n' POKE: Hex to Decimal Converter; String Array @(A) Memory Locator; Instructions on how to add a Printer; Bubble Sort; Camel; Memory Map; WUMPUS.
3. Mar. 80 Contains: Three Voice Music Assembler; Star Wars Music; Chopsticks; Chicago Loop; Lace Curtain; Character Set Size Multiplier; Rotation; National Distributor Info.
4. Apr/May 80 Contains: DMA Graphics (eliminates BOX & LINE commands, allows very complex graphics!); Reference Books, Product Review (Computer Ear-Speech Recognition Unit for Bally); Music Contest; RING; Alarm Clock; Byte Saving Hints.
5. June 80 Contains: Direct Color Video/Audio Circuit (for Color Monitor or VCR etc.); Galactibattle (Game); PRINT & STRING Tutorial; OTHELLO: ASCII Conversion Chart; Cursor Control Chart.
6. July 80 Contains: CURSOR Keyboard & 48K Memory Add-On info; ARCADE GOLF; Shell-Metzner Sort; Wavemakers "Maze-maker:; Floating Point Math; User Group Meetings.

VOL. 2, Issue 1 (August) Contains: Cursor Inventory Control (Business Software); "Connect Four" (a professional quality game); software for the Computer Ear (Anderson Research) - 'Fun With Music'.

VOL. 2, Issue 2 (Sept.) Contains: ASTROVISION ACQUIRES BALLY - an Editoral; Spider Web (Graphics Program); Reverse (Game); Design (a Graphics Program); Match Quiz (an Education Program); Note Match (an Education Program).

VOL. 2, Issue 3 (October) Contains: Peek n' Poke Tutorial; Critter - a Machine Language Program that creates a non-blinking FAST moving character; Poor Mans Memory Expansion (a tutorial); Sideswipe (car driving game); We Three Kings of Orient Are (3 Voice Chord Music); Machine Language Graphics Tutorial.

VOL. 2, Issue 4 (Nov/Dec) Contains: "ZGRASS32" Add-under info; New game cartridges; Programming Contest; Complete Bally Game Cartridge description; Chess program; Software reviews.

Vol. 2, Issue 5 (Jan/Feb) Contains: Winter CES News; Bio-Rythyms Compatability Analysis program; Towers of Hanoi game; Executive Time Card Calculator program; How to display all 256 colors on screen at same time; 1's & 5's game program; Line Resequence program.

Vol. 2, Issue 6 (Mar) Contains: New Basic Cartridge-a review; Metric Converter program; Dirty Programming Tricks (helpful hints) and the following programs: Morse Code Simulator; Hamurabi; Treasure Island.

SEND $1.75 per back issue desired OR $9.75 for Volumn I and $9.75 for Volumn II to: CURSOR, P.O. Box 266, N. Hollywood, CA 91603

BELLIGERENT BALLY BLUES



No need to have a nervous breakdown!!! If your unit refuses to function properly, take two aspirins, sit back and relax -- send your unit to us. For $25. (plus $5. return postage) we'll put your troubles to sleep and tune-up your Bally. We will require your O.K. on major repairs. See the Editorial this issue for additional information

SEND YOUR UNIT TO: The Cursor Group, 1010A West Magnolia Blvd., Suite 201, Burbank, CA 91506.

ZGRASS LANGUAGE

BY

TOM MEEKS



EDITORS NOTE: To show our readers the ease which ZGRASS, (the system language utilized by the AstroVision Add-Under for the Bally Arcade) can be learned, we asked Tom Meeks who is a proud owner of the UV1 (hi-res $5000, version of the Add-Under) to write an article for us explaining the use of the language. Tom, needless to say is in love with the unit and plans to purvey ZGRASS 32 unit to educational institutions.

Actually, it's much harder to write about Zgrass objectively than to learn it in the first place. There is something about the Zgrass language that brings out the "G-O-l-l-y!" and "Gee-Whiz!" in me. So, while I'll do my best to be objective and cool in appraising the capabilities of Tom DeFanti's brainchild...The truth is that it's just too much fun to pick at with any fervor.

There are now two micro's implementing the Zgrass language with the Bally custom chipset. The most familiar of the two is the AstroVision "Add-under" for the Bally Professional Arcade. The other is a commercial unit for video producers and graphics artists requiring a higher resolution picture. The Datamax UV-1 has a screen resolution of 320 X 202, or double that of the AstroVision machine. The operating systems are virtually identical. Four of the Datamax machines are now in the field with full production just getting underway. I have had mine for just a little over 1 month and have been more than pleased with what Zgrass allows me to do. Since I do not consider myself a professional programmer (or even a good hacker) I have been even more impressed with the ease with which complex art and animation can be done.

To this point, most of my experimentation and application with Zgrass has been purely graphics oriented. This, coupled with the fact that the Zgrass library of commands is still incomplete in several areas, leads me to limit this evaluation to graphics and game development capabilities we can expect from the AstroVision Zgrass-32 when it arrives. I will say this much however. Those readers engaged in educational activities are going to be very impressed with the versatility of Zgrass for Computer Aided Instruction.

Graphic development is the Zgrass' playground. All of the extravagant claims you may have heard are true! First, Zgrass is easier to learn than Basic. Second, when they claim to have 'powerful graphics commands' they mean POWERFUL GRAPHICS COMMANDS! Not only can you build a character and make it walk across the screen, but, you can recall the character to the screen in many different ways. Your character can be squashed, stretched, shrunk, expanded, turned upside down and backwards, duplicated any number of times and even used as a pattern when you design your own quilt...all with simple commands resembling the familiar BOX command used in the Bally Basic Cartridge.

Since we are all familiar with the Gunfight game that is resident in the Bally Arcade, perhaps the best way to appreciate the scope of the Zgrass graphics capabilities would be to relate how it could be used to produce some of the Gunfight graphics. Before we do this, however, some time should be spent on the format of Zgrass MACROS, or programs. Line numbers are not usually used in a Zgrass program. Instead, the SKIP command is used to move forward or backward through a program. There is no RUN command. All that is necessary is to type in the program name. It may seem awkward as you read this, but quickly becomes familiar in use. A typical

program to make boxes of decreasing sizes in four colors would look like this:

```
TOM={A=5Ø
B=3
BOX Ø,Ø,A,A,B
IF (B=B-1)<Ø,B=3
IF (A=A-5)>5,SKIP -2}

TOM
```

Once the MACRO called TOM is entered, it can be recalled at any time by typing in the name as we did above, or within any other macro like this:

```
EXAMPLE={PRINT "THIS PROGRAM MAKES
BOXES."
TOM
PRINT "WE HOPE YOU LIKE THESE BOXES."}
```

In this case, the MACRO called EXAMPLE prints out a statement, calls the MACRO named TOM (Which draws the boxes on the screen.) and then prints out a final statement. This capability of a MACRO calling other MACROS helps to simplify and organize otherwise complex programs into small, manageable units.

Since Zgrass gives us the natural ability to break down long problems into small units, this is exactly the method we will use to duplicate the Gunfight program. In this issue we will deal with generating the bullets at the top of the screen, the cacti, the trees and the covered wagon. There are three basic steps common to generating, saving and displaying each of these figures on the screen. Step 1 involves drawing the figure using keyboard commands or a joystick drawing program such as Scribble built into the Bally. Step 2 involves saving the drawn figure off the screen and storing it into memory. This is done with the SNAP (short for "snapshot") command. The syntax of the SNAP command looks like this:

```
SNAP NAME,XCENTER,YCENTER,XSIZE,YSIZE
```

Step 3 is the process of recalling the stored figure onto the screen...anywhere and any number of times that we need it. Notice that all of the bullets, cacti and trees look exactly the same. The syntax of the DISPLAY command is this:

```
DISPLAY NAME,XCENTER,YCENTER,DISPLAYM
ODE
```

We use the NAME in the DISPLAY command that we used in our original SNAP. We will now draw a bullet, SNAP it off the screen and then write a short MACRO that will DISPLAY the bullet 12 times at the top of the screen.

```
BOX Ø,Ø,4,6,1
BOX Ø,4,1,1,1
 .THESE DRAW THE BULLET
SNAP BULLET,Ø,1,4,7
 .THIS SETS UP A SNAP.ARRAY CALLED
 'BULLET'
 .AND SAVES IT IN MEMORY.
 .NOW WE WILL WRITE A MACRO TO DISPLAY
 THE BULLETS.
PUTBULLETS={A=23
DISPLAY BULLET,A,37,1
DISPLAY BULLET,-A,37,1
IF (A=A+5) 5Ø,SKIP -2}

PUTBULLET
 .CALLS AND EXECUTES THE MACRO THAT
 DISPLAYS 6
 .BULLETS FOR EACH PLAYER AT THE TOP
 OF THE SCREEN.
```

Now, each time one of the players is shot and we want to reset the screen, we merely call the MACRO called PUTBULLET. As yet, we haven't explained the DISPLAYMODE used in the DISPLAY command. The DISPLAYMODE allows us to "plop" the SNAPNAME on the screen or put it on with one of the built-in XOR,OR,AND or special PRIORITY WRITE features. These will be discussed in detail in future issues. The point is that we can generate different ways using the DISPLAY command.

The cacti and trees are generated and saved in exactly the same way. Suppose we drew a cactus centered at X=-2Ø,Y=Ø and a tree centered at X=2Ø,Y=Ø. The cactus might be 18 pixels high and 7 pixels wide. The tree might be 19 pixels high and 8 pixels wide. The SNAP'S used to save them are:

```
SNAP CACTUS,-2Ø,Ø,7,18
SNAP TREE,2Ø,Ø,8,19
```

Notice that the SNAP's are the same width and height as the figure being saved. This only works for figures that remain stationary on the screen once displayed. Obviously, this applies to the bullets, the cacti and the trees. It does not apply to the covered wagon. Since it moves up and down, the wagon must be SNAPPED off with a 'buffer' zone that allows us to erase all of the previously

displayed wagon with the new 'offset' display. Suppose we want to move the wagon 2 pixels at a time as we go up and down. We should have a buffer of at least 3 pixels at the top and bottom of the wagon figure. Suppose, further, that the wagon we draw is 21 pixels wide and 28 pixels high. Our SNAP command would look like this:

```
SNAP WAGON,Ø,Ø,21,34
```

Our MACRO for putting the wagon on the screen and moving it up and down could be written:

```
MOVEWAGON={Z=-3Ø
DISPLAY WAGON,Ø,Z,Ø
IF (Z=Z+2)<2Ø,SKIP -1
DISPLAY WAGON,Ø,Z,Ø
IF (Z=Z-2)>-3Ø,SKIP -1
SKIP -5}
```

This is all that is necessary to put the wagon on the screen and move it up and down...exactly like the movement in the Gunfight game. Imagine trying to do that in Basic! Zgrass makes us LOOK like machine language programmers!

In the next installment, we will create the gunfighters and WALK them around on the screen.



DIAMOND

BY

RICHARD SONNENBLICK

EDITORS NOTE: This is a Video Art Program. We publish only the best of those we receive. Invariably such programs make extensive use of For-TO-Next loops and can be valuable to the budding programmer as a learning device!
The author, Richard Sonnenblick is a Junior High School student. Congratulations go to Richard for a very excellent program. I hope the adults reading this are sufficiently chagrined

```
 1 .*********
 2 .*DIAMOND*
 3 .*********
 4 .BY RICHARD SONNENBLICK
 5 CLEAR ;&(9)=-1;GOTO 1Ø
 6 B=42;RETURN
 7 B=28;RETURN
 8 B=21;RETURN
 9 B=16;RETURN
1Ø A=RND (5);IF A=1GOTO 1Ø
2Ø GOSUB 4+A
3Ø C=Ø;D=85;FC=RND (128)x2
4Ø FOR E=1TO B;C=C+A;D=D-A
5Ø BOX Ø,Ø,C,D,3;MU=A;NEXT E
6Ø A=RND (5);IF A=1GOTO 6Ø
7Ø GOSUB 4+A;C=85;D=Ø
8Ø FOR E=1TO B;C=C-A;D=D+A
9Ø BOX Ø,Ø,C,D,3;MU=A;NEXT E
1ØØ GOTO 1Ø
```

YAHTZEE

BY

BRUCE DE VRIES

Yahtzee is a dice game for 1 to 4 players. The left side of the screen is played so that you roll the highest score on ones, two's, three's, etc. The right side is played similar to a poker hand: Full House, straights, 3 and 4 of a kind.

After the first roll of the dice you have two more chances to improve your score. To start this process pull the trigger. To remove a dice pull the trigger again. You may remove as many dice as you want or keep those dice you want by passing over them. If you like your first roll and want to keep all the dice shown, pull

the trigger and do nothing until the arrow appears at the top of the screen. After the final roll you must enter a score on the playing card in one of the categories shown using the joystick and trigger. A category can't be reused. Each turn must be scored somewhere. To "zero" a category, move the arrow to the desired location and pull the trigger.

Scoring of Points:
1's to 6's: Score is total of dice with that number. Example: (4-2-4-4-6)=12 scored as 4's. If the total score of 1's thru 6's reaches 63 a 35 point bonus is given.

3 or 4 of a Kind: Score is total of all dice. Example: (6-6-6-1-5=24 or 1-1-1-1 6=10)

Full House: 2 of one kind and 3 of another. Score is 25 points. Example: (2-2-2-5-5)

Small Straight: 4 numbers in sequence score is 30 points. Example: (2-3-4-5-2)

Large Straight: 5 numbers in sequence score is 40 points. Example: (2-3-4-5-6)

Yahtzee: 5 of a kind score is 50 points.

Chance: Any combination score is total of all dice. Example: 2-2-1-5-3=13

```
1Ø CLEAR ;BC=9;INPUT "#PLAYER"V
2Ø CLEAR ;A=-1Ø;NT=Ø;FOR B=6TO 64;@(B)=-1;
   NEXT B;FOR P=2ØTO 65STEP 15;@(P)=Ø;NEXT
   P;FOR I=1TO 6;PRINT #1,I,"'S";NEXT I
3Ø PRINT "BONUS";CY=4Ø;CX=A;FOR I=3TO 4;PR
   INT #1,I," OF A KIND";CX=A;NEXT I;PRINT
   "FULL HOUSE
4Ø CX=A;PRINT "S. STRAIGHT";CX=A;PRINT
   "L. STRAIGHT";CX=A;PRINT "YAHTZEE";C
   X=A;PRINT "CHANCE
5Ø CY=-18;PRINT "PLAYER #",;CX=A;PRINT
   "TOTAL";FOR T=1TO 13;FOR P=1TO V;H=P
   x15
6Ø FOR X=-7ØTO 7ØSTEP 3Ø;BOX X,-34,19,1
   9,1;BOX X,-34,17,17,2;NEXT X;GOSUB 4
   7Ø
7Ø FOR D=1TO 5;@(D)=Ø;NEXT D;R=Ø;FOR D=
   66TO 71;@(D)=Ø;NEXT D
8Ø R=R+1;FOR D=1TO 5;IF @(D)=Ø@(D)=RND
   (6);GOSUB 42Ø
9Ø NEXT D;F=Ø
1ØØ IF R=3GOTO 15Ø
11Ø IF TR(P)=ØGOTO 11Ø
12Ø IF TR(P)GOTO 12Ø
13Ø FOR D=1TO 5;BOX -1ØØ+Dx3Ø,-34,17,17,
   3;NT=3;FOR S=1TO 1ØØ;IF TR(P)F=1;BOX
    -1ØØ+Dx3Ø,-34,17,17,1;MU=49;NT=Ø;@(
   D)=Ø
14Ø NEXT S;BOX -1ØØ+Dx3Ø,-34,17,17,3;NEX
   T D;IF FGOTO 8Ø
15Ø U=Ø;FOR D=1TO 5;U=U+@(D);@(@(D)+65)=
   @(@(D)+65)+1;NEXT D;K=Ø;FOR I=66TO 7
   1;IF @(I)>KςK=@(I)
16Ø NEXT I;FOR I=1TO 5;FOR J=1TO 4;IF @(
   J)>@(J+1)Z=@(J);@(J)=@(J+1);@(J+1)=Z
17Ø NEXT J;NEXT I;C=1;R=1;S=Ø;W=Ø;FOR D=
   1TO 4;IF @(D)=@(D+1)GOTO 2ØØ
18Ø IF @(D)+1#@(D+1)W=W÷3x3;GOTO 2ØØ
19Ø W=W+1
2ØØ NEXT D;NT=3
21Ø CX=91xR-124;CY=48-Cx8;PRINT "←",
22Ø FOR D=1TO 5Ø;NEXT D;BOX CX-7,CY,5,7,
   2;IF TR(P)GOTO 27Ø
23Ø NT=Ø;IF JX(P)R=2-R÷2
24Ø C=C-JY(P);IF C=ØC=1
25Ø Z=5+R;IF C>ZςC=Z
26Ø GOTO 21Ø
27Ø IF R=2GOTO 32Ø
28Ø IF @(H+C-1Ø)>-1GOTO 22Ø
29Ø S=@(C+65)xC;IF @(H-3)>ØGOTO 4ØØ
3ØØ @(H-3)=@(H-3)-S;IF @(H-3)<-63@(H-3)=
   35;@(H+5)=@(H+5)+35;GOSUB 49Ø
31Ø GOTO 4ØØ
32Ø IF @(H+C-3)>-1GOTO 22Ø
33Ø IF C=1IF K>2S=U
34Ø IF C=2IF K>3S=U
35Ø IF C=3IF @(1)=@(2)IF @(4)=@(5)IF K=3
   S=25
36Ø IF C=4IF W>2S=3Ø
37Ø IF C=5IF W>3S=4Ø
38Ø IF C=6IF K=5S=5Ø;GOSUB 49Ø
39Ø IF C=7S=U
4ØØ @(H+Rx7+C-17)=S;@(H+5)=@(H+5)+S;GOSU
   B 47Ø;NEXT P;NEXT T
41Ø CLEAR ;FOR P=1TO V;PRINT "PLAYER",#3,
   P,#7,@(Px15+5);NEXT P;STOP
42Ø X=-1ØØ+Dx3Ø;Z=@(D);Q=Z÷2;IF RMBOX X,-
   34,3,3,1
43Ø IF Z>1BOX X-6,-28,3,3,1;BOX X+6,-4Ø,3
   ,3,1
44Ø IF Z>3BOX X+6,-28,3,3,1;BOX X-6,-4Ø,3
   ,3,1
45Ø IF Z>5BOX X+6,-34,3,3,1;BOX X-6,-34,3
   ,3,1
46Ø RETURN
47Ø FOR A=1TO 7;CY=48-Ax8;FOR D=1TO 2;CX=
   91xD-118;BOX CX+3,CY,13,7,2;B=@(H+Dx7
   +A-17);IF B>-1PRINT #2,B,
48Ø NEXT D;NEXT A;CX=-18;CY=-18;PRINT #1,
   P,;CX=4Ø;PRINT #3,@(Px15+5),;RETURN
49Ø FOR I=1TO 6;NT=9;MU=8Ø;NT=Ø;NEXT I;RE
   TURN
```

MATCH

BY

ED GROEBE

The object of this board game, for either one or two players, is to select the pairs of cards (A to T) which have matching numbers (0 to 9). As letter guesses are entered on the key-pad the cards are "turned over" to reveal the numbers. If a match is made the two cards are removed from the board and the player tries two more cards. If a match is not made the cards are turned down again.

If there are two players, they take turns guessing two cards at a time. At the end of the game the number of pairs matched for each player is shown. If there is just one player the number of tries needed to complete all 10 matches is shown at the end. The challenge is to have as few tries as possible--10 would be a perfect score.

Both the color and a number of the right of the screen indicate which player should enter a letter (blue is #1, red is #2). A short instruction at the beginning explains what to do.

```
 1Ø CLEAR ;BC=56;FC=55
 14 CY=Ø;CX=-3Ø;PRINT "MζAζTζCζH
 15 FOR C=1TO 5ØØ;NEXT C;CLEAR
 2Ø CY=Ø;PRINT "ζζζζζζζTRY TO MATCH";PRIN
    T ;PRINT "ζζζζPAIRS OF NUMBERS !
 3Ø PRINT ;PRINT ;PRINT "ζζζζζζKEY IN LET
    TERS";PRINT ;PRINT "ζζζζFOR YOUR GUES
    SES !
 32 PRINT ;PRINT ;PRINT "ζζζζζζ1 OR 2 PL
    AYERS ?
 34 P=KP
 37 IF P>5ØGOTO 34
 38 IF P<49GOTO 34
 39 CX=Ø;TV=P;Q=P-48
 4Ø PRINT
 6Ø K=Ø;J=Ø;S=Ø;T=Ø;CLEAR
 8Ø CY=Ø;CX=-15;PRINT "WAIT !
1ØØ FOR Z=ØTO 85;@(Z)=Ø;NEXT Z
13Ø U=RND (1Ø)-1
15Ø IF @(U)=1GOTO 13Ø
16Ø @(U)=1;GOTO 17Ø
17Ø FOR Z=1TO 2
175 W=RND (2Ø)
18Ø IF @(64+W)>ØGOTO 175
185 @(64+W)=U
19Ø NEXT Z
192 FOR Z=ØTO 9;IF @(Z)=ØGOTO 13Ø
196 NEXT Z;CLEAR
2ØØ X=-6Ø;Y=3Ø;Z=64
21Ø FOR Y=3ØTO -3ØSTEP -2Ø
22Ø FOR X=-6ØTO 6ØSTEP 3Ø
23Ø Z=Z+1;CX=X;CY=Y;TV=Z
25Ø BOX X,Y,2Ø,18,3
26Ø NEXT X;NEXT Y
3ØØ FOR P=1TO Q
3Ø2 CY=Ø;CX=75;PRINT #Ø,P
3Ø5 A=Ø
31Ø IF P=1T=T+1;BC=8;FC=7
311 IF P=2S=S+1;BC=8Ø;FC=7
32Ø W=KP
325 IF W<65GOTO 32Ø
326 IF W>84GOTO 32Ø
327 IF @(W)=1ØGOTO 32Ø
33Ø GOSUB 4ØØ+W
34Ø A=2Ø
35Ø V=KP
355 IF V<65GOTO 35Ø
356 IF V>84GOTO 35Ø
357 IF V=WGOTO 35Ø
358 IF @(V)=1ØGOTO 35Ø
36Ø GOSUB 4ØØ+V
37Ø GOTO 6ØØ
```

```
465 X=-6Ø;Y=3Ø;GOTO 49Ø
466 X=-3Ø;Y=3Ø;GOTO 49Ø
467 X=Ø;Y=3Ø;GOTO 49Ø
468 X=3Ø;Y=3Ø;GOTO 49Ø
469 X=6Ø;Y=3Ø;GOTO 49Ø
47Ø X=-6Ø;Y=1Ø;GOTO 49Ø
471 X=-3Ø;Y=1Ø;GOTO 49Ø
472 X=Ø;Y=1Ø;GOTO 49Ø
473 X=3Ø;Y=1Ø;GOTO 49Ø
474 X=6Ø;Y=10;GOTO 49Ø
475 X=-6Ø;Y=-1Ø;GOTO 49Ø
476 X=-3Ø;Y=-1Ø;GOTO 49Ø
477 X=Ø;Y=-1Ø;GOTO 49Ø
478 X=3Ø;Y=-1Ø;GOTO 49Ø
479 X=6Ø;Y=-1Ø;GOTO 49Ø
48Ø X=-6Ø;Y=-3Ø;GOTO 49Ø
481 X=-3Ø;Y=-3Ø;GOTO 49Ø
482 X=Ø;Y=-3Ø;GOTO 49Ø
483 X=3Ø;Y=-3Ø;GOTO 49Ø
484 X=6Ø;Y=-3Ø;GOTO 49Ø
49Ø GOTO 5ØØ+A
5ØØ BOX X,Y,18,16,2
51Ø CX=X;CY=Y;PRINT #Ø,@(W)
517 RETURN
52Ø BOX X,Y,18,16,2
525 CX=X;CY=Y;PRINT #Ø,@(V)
54Ø RETURN
6ØØ IF @(W)=@(V)GOTO 7ØØ
62Ø IF @(W)#@(V)GOTO 8ØØ
7ØØ @(W)=1Ø;@(V)=1Ø
71Ø IF P=1 K=K+1
711 IF P=2 J=J+1
72Ø IF K+J=1ØGOTO 9ØØ
73Ø A=45Ø
74Ø GOSUB 4ØØ+W
75Ø GOSUB 4ØØ+V
79Ø GOTO 3Ø5
8ØØ A=35Ø
82Ø GOSUB 4ØØ+W
825 A=36Ø
83Ø GOSUB 4ØØ+V
835 NEXT P
84Ø GOTO 3ØØ
85Ø BOX X,Y,2Ø,18,2
855 CX=X;CY=Y;TV=W;GOTO 88Ø
86Ø BOX X,Y,2Ø,18,2
865 CX=X;CY=Y;TV=V
88Ø BOX X,Y,2Ø,18,3;RETURN
9ØØ CLEAR
92Ø CY=Ø;CX=-24
93Ø PRINT "ζζζGAMEζζOVER
931 IF Q=2GOTO 97Ø
935 PRINT ;PRINT ;PRINT "ζζζζζζζ",T,"ζζTR
    YS !
94Ø GOTO 15
95Ø BOX X,Y,2Ø,2Ø,2
96Ø RETURN
97Ø PRINT ;PRINT ;PRINT "ζζ#1ζζMATCHED",K
98Ø PRINT "ζζ#2ζζζζζζζζζ",J
99Ø GOTO 15
```

HAND CONTROL CABLES

Have gremlins been munching on your hand control cables??? Why not replace those worn out cables with brand new ones from The Cursor Group? A set of 2 new cables $6.99 or order two sets (4 cables) for only $10.49. C.O.D. orders accepted!! CA residents add 6% tax.

SOUND PORT STUDY

BY

MIKE PEACE

EDITORS NOTE: The only methods of sound creation mentioned in CURSOR have described the use of PRINT statements, MU or the excellent 3 VOICE MUSIC ASSEMBLER in the March 1980 issue. Unfortunately, the use of the sound ports have never been adequately described. Manipulation of a sound port is by far the most dynamic and creative way to program almost any immaginable sound!

Many thanks to Mike Peace of WAVEMAKERS for explaining a complex subject in such a way that enables all of us to use these ports. F.C.

So, you want to create 3 Voice Music or great sound effects utilizing the sound ports???

Before we begin, try this sobering thought on for size; The Bally has 8 ports dealing with sound (&(16)through &(23)). Each one of these ports can have 256 different values (Ø to 255). By setting each of these ports to different values, we can come up with a multitude of different and unique sounds! In fact, there are approximately 256^8 possibilities. 256^4 would be 4,294,967,296; and that's only half way through. Scary huh?

Well, now that you are discouraged and don't feel like reading any further, let me tell you that it is far easier than it appears!!!

Think of the Bally as a recorder with 2 microphones. We are going to try to record 4 voices on them (2 on a mike). Let's give these people names:

&(17) is ALICE (Voice A)
&(18) is BONNIE (Voice B)
&(19) is CLYDE (Voice C)
&(23) is ANIMAL (White Noise) He can't sing a lick!
&(21) is VOLUME control for CLYDE &(19) and ANIMAL &(23)
&(22) is VOLUME control for ALICE &(17) and BONNIE &(18)
&(2Ø) is VIBRATO (Warble)
&(16) is SPEED (Master Oscillator)

CLYDE can sing as well as ALICE and BONNIE as long as ANIMAL is not trying to sing. ANIMAL is so noisy, that he can drown everyone out.

O.K. that's our story. Now, let's put everyone to work!

Our volumes can be set from Ø TO 255 and anyplace in between by typing in &(22)= 1ØØ; &(21) =12Ø or whatever (Ø is OFF-255 is FULL BLAST).

Now, we tell everyone what to sing (Ø is OFF or so high we can't hear it, 255 is somewhat subsonic, but you can hear it.) The real notes are in between Ø-255.

To demonstrate, lets tell ALICE to sing note 3Ø; BONNIE sings note 4Ø; CLYDE sings note 5Ø, and it's beautiful music together. Play with volumes if you like, merely type the values in.

```
&(17)=3Ø;&(18)=4Ø;&(19)=5Ø
```

&(16) is the SPEED of our recorder. 255 is the slowest, 1 is the fastest. Try changing the SPEED and all of our singers voices will change together.

We haven't heard from ANIMAL yet (good thing). But let's bring him in to see what he can do. Set his value to some figure between 1 and 255 and just see how he messes up our singers. He does some good impressions of explosions, but a singer he's not! Shut him off so we can hear what we're doing.

Try adding some VIBRATO. This does not always work with all three voices at the same time, but &(2Ø) set to any number makes for some interesting effects. The smaller numbers (1 to 15) are the fastest and larger numbers slow things down as we get closer to 255.

I have written a program which uses Hand Control #1, that I use to find exactly the note or sound I'm looking for. With it, you can experiment to find just the sound you want.

Move pointer with JY(1). Pull trigger and turn knob to set the value. Push JX(1) left or right to fine tune the value. The computer automatically sets &(16) SPEED to approximately 72 every time the NT is greater than Ø.

```
  1 .SOUND PORT STUDY
  2 .BY MIKE PEACE
 1Ø GOTO 2Ø
 11 PRINT "ζζζSPEED",;RETURN
 12 PRINT "ζζζALICE",;RETURN
 13 PRINT "ζζBONNIE",;RETURN
 14 PRINT "ζζζCLYDE",;RETURN
 15 PRINT "VIBRATOζ",;RETURN
 16 PRINT "VOLUME 2",;RETURN
 17 PRINT "VOLUME 1",;RETURN
 18 PRINT "ζζANIMAL",;RETURN
 2Ø CLEAR ;B=15;NT=Ø;FOR A=36TO -36STEP -
    1Ø;B=B+1
 3Ø CY=A;CX=-65
 35 GOSUB B-5
 4Ø PRINT #Ø,"ζ&(",B,")=
 5Ø NEXT A;A=5
 6Ø CX=32;A=A+JY(1)x1Ø;IF A<-3ØA=-35
 7Ø IF A>3ØA=35
 8Ø CY=A;B=(A+35)÷1Ø
 9Ø TV=95;TV=31;TV=31
 95 IF JX(1)@(B)=@(B)+JX(1);C=@(B);PRINT
    #7,C,;&(23-B)=C;GOTO 6Ø
1ØØ IF TR(1)@(B)=&(28);PRINT #7,@(B);&(23
    -B)=@(B)
11Ø GOTO 6Ø
```

BINDERS

NEVER MORE TO BE SEEN!! The Deluxe "CURSOR GROUP" 3-Ring Binder. This is your last chance to buy one of these binders, complete with CURSOR logo, made to hold all of your past issues of Cursor.

LIMITED SUPPLY --- so send us your order today. $6.99 includes postage and handling. CA residents add 6% sales tax. Sorry - no C.O.D.'s.

MANUALS ON SPECIAL

#3-Dis. Tiny Basic
#6-System Description
#7-Dis. System Software

A $30.97 value on sale for $25.99. For a complete description of these manuals see the listing under MANUALS*MANUALS* MANUALS. This "Special Price" ends on June 1, 1981. C.O.D. orders accepted.

WAVEMAKERS

" 2 FOR 1 SALE "

WAVEMAKERS ANNOUNCES THAT DUE TO RISING COSTS WE REGRET THAT WE WILL HAVE TO RAISE OUR PRICES ONCE AGAIN. BUT WE'VE COME UP WITH AN IDEA TO SAVE YOU OUR CUSTOMER SOME MONEY. ALL ORDERS POSTMARK BEFORE JUNE 1 (NO RE-ORDERS OR DISTRIBUTERS) CAN TAKE ADVANTAGE OF OUR 2 FOR 1 SALE. JUST PICK TWO TAPES NUMBER 1 THROUGH 6, WE'LL COMBINE THEM BOTH ON ONE TAPE FOR THE PRICE OF ONE TAPE. EXCLUDING GUITAR COURSE AND BACKGAMMON.
TAPE 1 THROUGH 6 $10.95 TAPE #7 GUITAR COURSE $19.95 TAPE #8 BACKGAMMON $15.95

WAVEMAKERS BOX 94801 SCHAUMBURG, IL 60193



TAPE 1- Max & Horse Race
TAPE 2- Clue & Flying Ace
TAPE 3- Maze Race and Space Chase
TAPE 4- Slot Machine and Pervision
TAPE 5- Music Composer & Yahtzee
TAPE 6- Mouse in the Hat & Speed Math



STAR TREK

You are at the helm of the 400,000 metric ton star ship looking through the front viewscreen into deep space. It is a weary ship. You have already done battle with a class I cruiser and destroyed a Klingon outpost. Suddenly, a siren sounds "Red Alert!" Klingon Battle Cruiser' flashes across the sensor readout as the vessel appears on screen, very small at first, but getting larger. He's attacking at warp factor 6! Before you reset the enemy fires! An alarm sounds! Engineer Ing reports the hit damaged the photon torpedos & the forward shield is still weakened from the last battle. Another enemy blast fills the screen as it's about to hit. You must act quickly! You are not yet in phaser range. What will you do? STAR TREK, the aggressive game from Esoterica Ltd.

MINI GOLF

A game of computer putt-putt for one to four players. Use the joystick to line up your shot. Hit the ball hard or soft or with the perfect "Golfer's Touch." Zeak it off the side boards for those "freaky" shots but don't hit it too hard or you'll end up in the rough. Computer keeps running score for one to four players, including penalty stroke. It also prints player up and hole no.

******

Both programs make full use of color graphic and sound effects, and perform more functions that can be listed here. Send check or money order to:

Esoterica Ltd.
3328 Red Fox Run
Warren, Ohio 44485

L&M SOFTWARE
8599 FRAMEWOOD DR.
NEWBURGH, TN 47630
Visa & Master Chg. Welcome

Write for Free Catalog



THE BLACK LAGOON

SPACE QUEST 2001



THE BLACK LAGOON: You are the pilot of a boat, which must reach the dock on the other side of the Black Lagoon by passing thru the interconnecting waterways and avoiding the moster.

SPACE QUEST 2001: Your interstellar space craf must save a friendly colony, on the planet, fro the Star Plunderer. You must break thru the forcefield and destroy the enemy, using lasers and missiles.

$10.00 for back programs

CLASSIC

P. O. BOX 266, NO. HOLLYWOOD, CA. 91603

FIRST CLASS